
The I/O relation
A worksheet motivated by a parameters estimation problem

restart;
with (DifferentialAlgebra0);

A compartmental model x1 is observed while x2 is not.
params := [k[e], V[e], k[12], k[21]];

R := DifferentialRing (derivations = [t], 
                       blocks = [[x[1],x[2]], params], 
                       parameters = params,
                       notation = diff);

edoA := diff(x[1](t),t) = -k[12]*x[1](t) + k[21]*x[2](t) - 
                               (V[e]*x[1](t))/(k[e]+x[1](t));
edoB := diff(x[2](t),t) = k[12]*x[1](t) - k[21]*x[2](t);

The perfect differential ideal generated by the equations, saturated by the multiplicative family 
generated by the denominators.
The base field is the field of the rational fractions generated by the parameters over Q.
A sophisticated way to express the fact that we do not want to discuss the possible vanishing of 
parameters.

ideal := RosenfeldGroebner ([edoA,edoB], basefield=field
(generators=params), R);

Equations (ideal [1], solved);

Change of ranking: eliminate the non-observed variable x2, in order to get an ODE constraining x1, its 
derivates and the parameters

IOideal := Pardi (ideal [1], 
                  ranking (derivations = [t], 
                            blocks = [x[2], x[1], params]));

The sought equation

IOrel := Equations (IOideal, leader=derivative(x[1](t)))[1];

A work in progress with F. Lemaire, M. Rosenkranz and Georg Regensburger ...
The RatBilge algorithm should help transforming differential equations into integral equations.

normalized_IOrel := IOrel / Initial (IOrel, R):

L := RatBilge (normalized_IOrel, t, R);

The meaning of the above output
L[1] + add (Diff (L[i], t$(i-1)), i = 2 .. nops (L));

Just check
normal (normalized_IOrel - add (Differentiate (L[i], t$(i-1), 
R), i = 1 .. nops (L)));

